
HP-35s Calculator Program XYZ   2

Convert XYZ Geocentric Co-ordinates on any ellipsoid to
Latitude, Longitude and Height

Programmer: Dr. Bill Hazelton

Date: October, 2007. Mnemonic: L for ‘XYZ to Lat/Long’

Line Instruction Display User Instructions
L001 LBL L Press  XEQ  L  ENTER to run program
L002 CLSTK
L003 FS?  10
L004 GTO  L008
L005 SF  1
L006 SF  10
L007 GTO  L009
L008 CF  1
L009 XYZ TO LATLONG
L010 PSE
L011 6378137 Value of a for WGS84/NAD83/GRS80
L012 STO  A
L013 0.006694381 Value of e2 for WGS84/NAD83/GRS80
L014 STO  E
L015 SEMIMAJOR AXIS
L016 PSE
L017 INPUT  A 6378137 Enter value of a if different;  Press R/S
L018 E SQUARED
L019 PSE
L020 INPUT  E 0.006694381 Enter value of e2 if different;  Press R/S
L021 ENTER X CO-ORD
L022 PSE
L023 INPUT  X Enter X co-ordinate of point;  Press R/S
L024 ENTER Y CO-ORD
L025 PSE
L026 INPUT  Y Enter Y co-ordinate of point;  Press R/S
L027 ENTER Z CO-ORD
L028 PSE
L029 INPUT  Z Enter Z co-ordinate of point;  Press R/S
L030 RCL  Y
L031 RCL  X
L032 ÷
L033 ATAN
L034 STO  L
L035 1
L036 RCL–  E
L037 RCL  A
L038 x2

Line Instruction Line Instruction Line Instruction
L039 × L084 RCL  A L129 RCL–  Z
L040 √x L085 1 L130 x2

L041 STO  B L086 RCL  F L131 STO+  H
L042 RCL  A L087 SIN L132 RCL  X
L043 x2 L088 x2 L133 x2

L044 RCL  B L089 RCL×  E L134 RCL  Y
L045 x2 L090 – L135 x2

L046 – L091 √x L136 +
L047 RCL  B L092 ÷ L137 RCL  Z
L048 x2 L093 STO  V L138 x2

L049 ÷ L094 RCL  F L139 +
L050 STO  D L095 COS L140 √x
L051 RCL  X L096 × L141 RCL  G
L052 x2 L097 RCL  L L142 √x
L053 RCL  Y L098 COS L143 –
L054 x2 L099 × L144 ENTER
L055 + L100 STO  C L145 ABS
L056 √x L101 RCL–  X L146 ÷
L057 STO  P L102 x2 L147 RCL  H
L058 RCL  Z L103 STO  H L148 √x
L059 x < > y L104 RCL  C L149 ×
L060 ÷ L105 x2 L150 STO  H
L061 RCL  A L106 STO  G L151 RCL  F
L062 RCL÷  B L107 RCL  C L152 →HMS
L063 × L108 RCL  L L153 STO  F
L064 ATAN L109 TAN L154 LATITUDE
L065 STO  U L110 × L155 PSE
L066 SIN L111 STO  C L156 VIEW  F
L067 3 L112 x2 L157 RCL  L
L068 yx L113 STO+  G L158 →HMS
L069 RCL×  B L114 RCL  C L159 STO  L
L070 RCL×  D L115 RCL–  Y L160 LONGITUDE
L071 RCL+  Z L116 x2 L161 PSE
L072 RCL  U L117 STO+  H L162 VIEW  L
L073 COS L118 RCL  V L163 ELLIPSE HEIGHT
L074 3 L019 1 L164 PSE
L075 yx L120 RCL–  E L165 VIEW  H
L076 RCL×  A L121 × L166 RCL  H
L077 RCL×  E L122 RCL  F L167 RCL  L
L078 RCL  P L123 SIN L168 RCL  F
L079 x < > y L124 × L169 FS?  1
L080 – L125 STO  C L170 CF  10
L081 ÷ L126 x2 L171 RTN
L082 ATAN L127 STO+  G
L083 STO  F L128 RCL  C

Notes

(1) A program to convert X, Y, Z geocentric co-ordinates to latitude, longitude and
ellipsoidal height on any ellipsoid.

(2) The assumption is that the distances are in meters, but by using feet for the semi-
major axis of the ellipsoid, co-ordinates in feet can be produced.

(3) The program pre-enters the parameters for the WGS84/NAD83/GRS80 ellipsoid
by default (in meters), to save you having to remember these. If you want a
different ellipsoid, enter the appropriate a and e2 values at the prompt (A and E).
To use the provided values, just press R/S when they appear.

(4) The resulting latitude, longitude and height are displayed with a prompt or label.
Note that the program does not clear registers after use. You can get ν for the
point by using the RCL V keystrokes.

(5) The latitude and longitude are displayed in HP notation, i.e., DDD.MMSS. The
height is assumed to be in the same units as the semi-major axis, by default,
meters.

(6) The sign convention with latitudes, longitudes and heights is the standard one, as
follows. Latitudes in the southern hemisphere are negative. Longitudes west of
Greenwich are negative, i.e., all longitudes in the US are negative. Heights below
the ellipsoid are shown as negative.

(7) Pay particular attention to the sign of the co-ordinate values for the point. West
of longitude 90° W, all X values will be negative; west of Greenwich (i.e., all of
the US) all Y values are negative; south of the equator, all Z values will be
negative. Incorrect signs will throw the position out rather dramatically.

(8) Owing to rounding in the calculator, it is possible for a value like 80° 00' 00" to
be displayed as 79.5960, rather than 80.0000. You can convert the results to the
appropriate representation in your head or on paper, as there is no difference in
the internal calculations or the results.

(9) When the program finishes, it places the computed values on the stack. The
height of the point above the ellipsoid is in the Z stack register. The longitude (in
HP notation) is in the Y stack register. The latitude (in HP notation) is in the X
stack register. This allows the program to be called from another program (as a
sub-routine) and return values on the stack for further processing.

(10) The program sets Flag 10, to allow prompts to be displayed. When the program
finishes, it sets Flag 10 back to its original value. To do this, the program uses
Flag 1.

Theory

The program implements the following equations:

� 

λ = arctan Y
X

⎛ 
⎝ 
⎜ 

⎞ 
⎠ 
⎟ [1]

� 

b2 = a2(1 − e2) [the semi-minor axis length] [2]

� 

p = X2 + Y2 [3]

� 

tan u = Z
p

⎛ 
⎝ ⎜ 

⎞ 
⎠ ⎟ 

a
b

⎛ 
⎝ ⎜ 

⎞ 
⎠ ⎟ [4]

ν  =  

� 

a
1 − e2 sin2 φ

[5]

� 

φ = arctan Z + εb sin3 u
p − e2a cos3 u
⎛ 

⎝ 
⎜ 

⎞ 

⎠ 
⎟ [6]

� 

ε = e'2 = (a2 − b2)
b2

[the second eccentricity] [7]

X  =  ν  cos φ  cos λ [8]

Y  =  ν  cos φ  sin λ [9]

Z  =  ν (1 – e2)  sin φ           [10]

Equation [1] provides a direct solution for the longitude, λ. The program then computes a variety
of intermediate results, before using equation [6] to compute the latitude. This is a direct solution.
The program then computes the X, Y, Z location for the point on the ellipsoid at φ, λ, using
equations [8] to [10]. The distance between this point and the given X, Y Z co-ordinates is
computed to determine h, the ellipsoidal height.

The distances to the Earth’s center from the original X, Y, Z co-ordinates and the point on the
ellipsoid are computed to get the correct sign for h, i.e., is the point above or below the surface of
the ellipsoid.

The equations are from Bowering (1976).

Sample Computation

Inputs a  =  6 378 137 m
e2  =  0.006 694 381 (WGS84/NAD83/GRS80 parameters)
X  =   1 353 776.483 m
Y  = –5 052 362.616 m
Z  =   3 637 981.622 m

Results φ  =  35° 00' 00" N   (displayed as F = 35.000000)
λ  =  75° 00' 00" W  (displayed as L = –75.000000)
h  =  200.000 m        (displayed as H = 200.0000049)

Note that the precision of the answer depends upon the precision of the input. One millimeter
(0.001m) at the surface of the ellipsoid equates to 0.00003" of arc of latitude and generally a
smaller amount of longitude. Consider your input precisions and adjust the quoted precision of
the outputs to match. You can use the SHOW function to see all the digits in a number.

Running the Program
Begin by pressing  XEQ  L  ENTER

The calculator displays  XYZ to LATLONG, briefly, then displays SEMIMAJOR AXIS, briefly.

The calculator displays: A?
6,378,137.0000

This is the NAD83/WGS84/GRS80 ellipsoid semi-major axis. If this is OK, press R/S; if not key
in correct value and press R/S.

The calculator displays: E  SQUARED, briefly.
E?
0.006694381    (suitably rounded, according to your settings)

This is the eccentricity of the NAD83/WGS84/GRS80 ellipsoid, e2. If this OK, press R/S; if not,
key in correct value and press R/S.

The calculator displays: ENTER X COORD, briefly.
X?

[Whatever value happens to be in this register]

Key in the X co-ordinate of the point and press R/S. Remember to use the appropriate sign, if
negative. In the given example, key in 1353776.483 and press R/S.

The calculator displays: ENTER Y COORD, briefly.
Y?

[Whatever value happens to be in this register]

Key in the Y co-ordinate of the point and press R/S. Remember to use the appropriate sign, if
negative. In the given example, key in –5052362.616 and press R/S.

The calculator displays: ENTER Z COORD, briefly.
Z?

[Whatever value happens to be in this register]

Key in the Z co-ordinate for the point and press R/S. Remember to use the appropriate sign, if
negative. In the given example, key in 3637981.622 and press R/S.

The calculator displays RUNNING for a short while.

The calculator displays: LATITUDE, briefly.
F=

35.00000000 Press R/S

The calculator display LONGITUDE, briefly.
L=

–75.0000000 Press R/S

The calculator displays: ELLIPSE HEIGHT, briefly.
H=

200.0000049 Press R/S

The calculator now completes its run, placing the latitude, longitude and height on the stack, and
resetting Flag 10 to its state when the program started.



HP-35s Calculator Program XYZ   2
XYZ Geocentric Co-ordinates to Lat/Long/Ht

- 6 -

These calculations agree with the NGS website computations to within 0.001 m in height and
0.00002" in latitude and longitude.

Storage Registers Used

A Semi-major axis of the ellipsoid, a. By default, set to  6378137 m.

B Semi-minor axis of the ellipsoid. Computed from a and e2.

C Intermediate value.

D Second eccentricity of ellipsoid.

E Eccentricity of the ellipsoid, e2. By default, set to 0.006694381.

F Latitude (geodetic) of the point, φ.

G Distance from ellipsoid point to center of Earth.

H Ellipsoidal height of the point, h.

L Longitude of the point, λ.

P Intermediate value.

U Intermediate value.

V The radius of curvature of the ellipsoid in the prime vertical, ν.

X Geocentric X co-ordinate of the point.

Y Geocentric Y co-ordinate of the point.

Z Geocentric Z co-ordinate of the point.

Labels Used

Label L Length  =   643 Checksum  =  36FE

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly.
Use the sample computation to check proper operation after entry.

The NGS website for the interactive XYZ ⇔ lat/long/height converter:

http://www.ngs.noaa.gov/TOOLS/XYZ/xyz.shtml


